37/replication_package/Do/Figures/Fig8.do

**********
* Immigration and Redistribution
* Alesina, Miano, Stantcheva
* RESTUD

* Figure 8
**********

clear all

* Specify directory of the replication package
global dir "/Users/armandomiano/Dropbox/AMS_Redistribution/Data/Survey_Data/Replication RESTUD sharing"
cd "$dir"

* Declare output sub-directory

global outdir "Out/Figures"

* Load the data
use "Data/survey_analysis.dta", clear

* Countries included in the analysis
global countries "US UK DE FR IT SE"

* Generate flags - To ensure answers quality
gen flag_1=0
gen flag_2=0

bysort country treatment_recod: egen min_duration = pctile(duration), p(2)
bysort country treatment_recod: egen max_duration = pctile(duration), p(98)

replace flag_1=1 if duration<min_duration
replace flag_2=1 if duration>max_duration

* Drop respondents in bottom and top 2% of distribution of time spent on the survey
keep if flag_1==0 & flag_2==0

*Drop variable used to exclude inattentive respondents
drop flag_1 flag_2 min_duration max_duration

* Gen dummies for Categorical variables

* Immigration is not a problem
gen imm_not_problem=(q_imm_problem==16 | q_imm_problem==17)
replace imm_not_problem=. if q_imm_problem==.
la var imm_not_problem "The current number of immigrants is not a problem"
	
* When should immigrants be entitled to get benefits?
gen imm_benefits_soon=(q_imm_benefits==1 | q_imm_benefits==2 | q_imm_benefits==4)
replace imm_benefits_soon=. if q_imm_benefits==.
la var imm_benefits_soon "Immigrants should get benefits in less than 3 years"

* When should immigrants be allowed to apply for citizenship?
gen imm_citizenship_soon=(q_imm_citizenship==1 | q_imm_citizenship==2)
replace imm_citizenship_soon=. if q_imm_citizenship==.
la var imm_citizenship_soon "Immigrants should get citizenship in 2 or 5 years"

* When would you consider an immigrant "Truly American"?
gen trully_american_cit=(q_imm_american==7 | q_imm_american==1 | q_imm_american==4)
replace trully_american_cit=. if q_imm_american==.
la var trully_american_cit "Consider American at citizenship or sooner"

* Government should care about everyone
gen govt_everyone=(q_govt_imm==7 | q_govt_imm==6)
replace govt_everyone=. if q_govt_imm==.
la var govt_everyone "The government should care equally about everyone"


****************
* Gen controls *
****************


* Gen Left-right variables
* Based on vote or voting intensions
gen left=(party_voted==4 | party_voted==5)
replace left=. if party_voted==. | party_voted==6 | party_voted==0
gen right=(party_voted==1 | party_voted==2)
replace right=. if party_voted==. | party_voted==6 | party_voted==0
gen center=(party_voted==3)
replace center=. if party_voted==. | party_voted==6 | party_voted==0

label var left "Left-wing"
label var right "Right-wing"
label var center "Center"

foreach i in left right center{
label val `i' binary
}


* Gen variables for HI sectors, with low/high educ
gen sector_dummy_h=sector_dummy*university_degree
gen sector_dummy_l=sector_dummy*(1-university_degree)


*Young
gen young=(age<45)


gen US=(country=="US")
gen UK=(country=="UK")
gen IT=(country=="IT")
gen SE=(country=="SE")
gen DE=(country=="DE")
gen FR=(country=="FR")



* Keep control group only
****************
keep if control==1
****************


* Declare variables to be included in the graph 

global imm_vars imm_not_problem imm_benefits_soon imm_citizenship_soon ///
						trully_american_cit govt_everyone



********** BY COUNTRY **********

*** Means of variables ***
foreach x in US UK FR IT DE SE {	
foreach y in $imm_vars {
su `y' if country=="`x'"
local m`y'`x' = r(mean)
}
}


preserve
        clear
        set obs 26
        egen t = seq()
		label define var 5 "Govt. should care about everyone" 4 "American upon citiz. or before" 3 "Imm. allowed to get citiz. soon" ///
		2 "Imm. should get benefits soon" 1 "Imm. not a problem" 
		label values t var
		replace t=t/5
		
		
        gen m_US = .
		gen m_UK = .
        gen m_FR = .
        gen m_IT = .
		gen m_DE = .
        gen m_SE = .
		
		gen n=_n
		foreach x in US UK FR IT DE SE {	
		replace m_`x'=`mgovt_everyone`x'' if n==25
		replace m_`x'=`mtrully_american_cit`x'' if n==20
		replace m_`x'=`mimm_citizenship_soon`x'' if n==15
		replace m_`x'=`mimm_benefits_soon`x'' if n==10
		replace m_`x'=`mimm_not_problem`x'' if n==5
		}


	
		twoway 	(scatter t m_US, mcolor(red) msize(large) lcolor(midblue) lpattern(solid) msymbol(D)) ///
				(scatter t m_UK, mcolor(blue) msize(large) lcolor(red) lpattern(solid) msymbol(O)) ///
				(scatter t m_FR, mcolor(dkgreen) msize(large) lcolor(red) lpattern(solid) msymbol(S)) ///
				(scatter t m_IT, mcolor(orange) msize(large) lcolor(red) lpattern(solid) msymbol(T)) ///
				(scatter t m_DE, mcolor(balck) msize(large) lcolor(red) lpattern(solid) msymbol(C)) ///
				(scatter t m_SE, mcolor(purple) msize(vlarge) lcolor(midblue) lpattern(solid) msymbol(X)), ///
                ytitle("") ylabel(1 2 3 4 5, value labsize(medium) angle(0) glcolor(gs16) noticks)  ///
                legend(order(1 2 3 4 5 6) cols(3) label(1 "US") label(2 "UK") label(3 "France") label(4 "Italy") label(5 "Germany") label(6 "Sweden") size(2.3)) ///
				xtitle("Share Answering Yes") ///
				yline(1, lcolor(gs14) lpattern("-")) yline(2, lcolor(gs14) lpattern("-")) ///
				yline(3, lcolor(gs14) lpattern("-")) yline(4, lcolor(gs14) lpattern("-")) ///
				yline(5, lcolor(gs14) lpattern("-")) ///
                graphregion(color(white)) plotregion(color(white))
				graph export "$outdir/Figure_8_A.eps", as(eps) replace
				
restore		
		

********** BY CORE CHARACTERISTICS **********

gen var_1 = left
gen var_2 = right
gen var_3 = university_degree
gen var_4 = (1-university_degree)
gen var_5 = sector_dummy_l
gen var_6 = sector_dummy_h
gen var_7 = (1-sector_dummy)

*** Means of variables ***
forval i=1(1)7 {	
foreach y in $imm_vars {
su `y' if var_`i'==1
local `y'_m`i' = r(mean)
}
}


preserve
        clear
        set obs 26
        egen t = seq()
		label define var 5 "Govt. should care about everyone" 4 "American upon citiz. or before" 3 "Imm. allowed to get citiz. soon" ///
		2 "Imm. should get benefits soon" 1 "Imm. not a problem" 
		label values t var
		replace t=t/5
		
		forval i=1(1)7{
        gen m_`i' = .
		}
		
		*label define var 1 "" 2 "" 3 "" 4 "" 5 "" 

		gen n=_n
		
		forval i=1(1)7 {	
		replace m_`i'=`govt_everyone_m`i'' if n==25
		replace m_`i'=`trully_american_cit_m`i'' if n==20
		replace m_`i'=`imm_citizenship_soon_m`i'' if n==15
		replace m_`i'=`imm_benefits_soon_m`i'' if n==10
		replace m_`i'=`imm_not_problem_m`i'' if n==5
		}


	
		twoway 	(scatter t m_1, mcolor(eltblue) msize(large) lcolor(eltblue) lpattern(solid) msymbol(S)) ///
				(scatter t m_2, mcolor(maroon) msize(large) lcolor(maroon) lpattern(solid) msymbol(D)) ///
				(scatter t m_3, mcolor(dkgreen) msize(large) lcolor(dkgreen) lpattern(solid) msymbol(C)) ///
				(scatter t m_4, mcolor(yellow*1.3) msize(large) lcolor(yellow*1.3) lpattern(solid) msymbol(T)) ///
				(scatter t m_5, mcolor(orange) msize(vlarge) lcolor(orange) lpattern(solid) msymbol(+) mlwidth(medthick)) ///
				(scatter t m_6, mcolor(black) msize(huge) lcolor(black) lpattern(solid) msymbol(X) mlwidth(medthick)) ///
				(scatter t m_7, mcolor(navy) msize(vlarge) lcolor(navy) lpattern(solid) msymbol(pipe) mlwidth(medthick)) , ///
                ytitle("") ylabel(1 2 3 4 5, value labsize(medium) angle(0) glcolor(gs16) noticks)  ///
                legend(order(1 2 3 4 5 6 7) cols(3) label(1 "Left-Wing") label(2 "Right-Wing") label(3 "College") label(4 "No College") label(5 "H Imm, No college") label(6 "H Imm, College") label(7 "No H Imm") size(2.2)) ///
				xtitle("Share Answering Yes") xlabel(0(0.2)1) ///
				yline(1, lcolor(gs14) lpattern("-")) yline(2, lcolor(gs14) lpattern("-")) ///
				yline(3, lcolor(gs14) lpattern("-")) yline(4, lcolor(gs14) lpattern("-")) ///
				yline(5, lcolor(gs14) lpattern("-")) ///
				xlabel(0.1(0.2)0.7) ///
                graphregion(color(white)) plotregion(color(white))
				graph export "$outdir/Figure_8_B.eps", as(eps) replace
				
restore